


xdPixel

RAY MARCHING

RAY MARCHING 101 – PART 2

JANUARY 7, 2016 | ADMIN | LEAVE A COMMENT



URL: <http://glslsandbox.com/e#29798.0>

```
1 // Ray Marching Tutorial (With Shading)
2 // By: Brandon Fogerty
3 // bfogerty at gmail dot com
4 // xdpixel.com
5
6 // Ray Marching is a technique that is very similar to Ray Tracing.
```

```

7 // In both techniques, you cast a ray and try to see if the ray intersects
8 // with any geometry. Both techniques require that geometry in the scene
9 // be defined using mathematical formulas. However the techniques differ
10 // in how the geometry is defined mathematically. As for ray tracing,
11 // we have to define geometry using a formula that calculates the exact
12 // point of intersection. This will give us the best visual result however
13 // some types of geometry are very hard to define in this manner.
14 // Ray Marching using distance fields to describe geometry. This means all
15 // we need to know to define a kind of geometry is how to measure the distance
16 // from any arbitrary 3d position to a point on the geometry. We iterate or "march"
17 // along a ray until one of two things happen. Either we get a resulting distance
18 // that is really small which means we are pretty close to intersecting with some kind
19 // of geometry or we get a really huge distance which most likely means we aren't
20 // going to intersect with anything.
21
22 // Ray Marching is all about approximating our intersection point. We can take a pretty
23 // good guess as to where our intersection point should be by taking steps along a ray
24 // and asking "Are we there yet?". The benefit to using ray marching over ray tracing is
25 // that it is generally much easier to define geometry using distance fields rather than
26 // creating a formula to analytically find the intersection point. Also, ray marching makes
27 // certain effects like ambient occlusion almost free. It is a little more work to compute
28 // the normal for geometry. I will cover more advanced effects using ray marching in a later
29 // For now, we will simply ray march a scene that consists of a single sphere at the origin
30 // We will not bother performing any fancy shading to keep things simple for now.
31
32 #ifdef GL_ES
33 precision mediump float;
34 #endif
35
36 uniform vec2 resolution;
37
38 //-----
39 // The sphere function takes in a point along the ray
40 // we are marching and a radius. The sphere function
41 // will then return the distance from the input point p
42 // to the closest point on the sphere. The sphere is assumed
43 // to be centered on the origin which is (0,0,0).
44 float sphere( vec3 p, float radius )
45 {
46     return length( p ) - radius;
47 }
48
49 //-----
50 // The map function is the function that defines our scene.
51 // Here we can define the relationship between various objects
52 // in our scene. To keep things simple for now, we only have a single
53 // sphere in our scene.
54 float map( vec3 p )
55 {
56     return sphere( p, 3.0 );
57 }
58
59 //-----
60 // This function will return the normal of any point in the scene.
61 // This function is pretty expensive so if you need the normal, you should
62 // call this function once and store the result. Essentially the way it works
63 // is by offsetting the input point "p" along each axis and then determining the
64 // change in distance at each new point along each axis.
65 vec3 getNormal( vec3 p )
66 {
67     vec3 e = vec3( 0.001, 0.00, 0.00 );
68
69     float deltaX = map( p + e.xyy ) - map( p - e.xyy );
70     float deltaY = map( p + e.yxy ) - map( p - e.yxy );
71     float deltaZ = map( p + e.yyx ) - map( p - e.yyx );
72
73     return normalize( vec3( deltaX, deltaY, deltaZ ) );

```

```

74 }
75
76 //-----
77 // The trace function is our integration function.
78 // Given a starting point and a direction, the trace
79 // function will return the distance from a point on the ray
80 // to the closest point on an object in the scene. In order for
81 // the trace function to work properly, we need functions that
82 // describe how to calculate the distance from a point to a point
83 // on a geometric object. In this example, we have a sphere function
84 // which tells us the distance from a point to a point on the sphere.
85 float trace( vec3 origin, vec3 direction, out vec3 p )
86 {
87     float totalDistanceTraveled = 0.0;
88
89     // When ray marching, you need to determine how many times you
90     // want to step along your ray. The more steps you take, the better
91     // image quality you will have however it will also take longer to render.
92     // 32 steps is a pretty decent number. You can play with step count in
93     // other ray marchign examples to get an intuitive feel for how this
94     // will affect your final image render.
95     for( int i=0; i <32; ++i)
96     {
97         // Here we march along our ray and store the new point
98         // on the ray in the "p" variable.
99         p = origin + direction * totalDistanceTraveled;
100
101         // "distanceFromPointOnRayToClosestObjectInScene" is the
102         // distance traveled from our current position along
103         // our ray to the closest point on any object
104         // in our scene. Remember that we use "totalDistanceTraveled"
105         // to calculate the new point along our ray. We could just
106         // increment the "totalDistanceTraveled" by some fixed amount.
107         // However we can improve the performance of our shader by
108         // incrementing the "totalDistanceTraveled" by the distance
109         // returned by our map function. This works because our map function
110         // simply returns the distance from some arbitrary point "p" to the closest
111         // point on any geometric object in our scene. We know we are probably about
112         // to intersect with an object in the scene if the resulting distance is very small.
113         float distanceFromPointOnRayToClosestObjectInScene = map( p );
114         totalDistanceTraveled += distanceFromPointOnRayToClosestObjectInScene;
115
116         // If our last step was very small, that means we are probably very close to
117         // intersecting an object in our scene. Therefore we can improve our performance
118         // by just pretending that we hit the object and exiting early.
119         if( distanceFromPointOnRayToClosestObjectInScene < 0.0001 )
120         {
121             break;
122         }
123
124         // If on the other hand our totalDistanceTraveled is a really huge distance,
125         // we are probably marching along a ray pointing to empty space. Again,
126         // to improve performance, we should just exit early. We really only want
127         // the trace function to tell us how far we have to march along our ray
128         // to intersect with some geometry. In this case we won't intersect with any
129         // geometry so we will set our totalDistanceTraveled to 0.00.
130         if( totalDistanceTraveled > 10000.0 )
131         {
132             totalDistanceTraveled = 0.0000;
133             break;
134         }
135     }
136
137     return totalDistanceTraveled;
138 }
139
140 //-----

```

```

141 // Standard Blinn lighting model.
142 // This model computes the diffuse and specular components of the final surface color.
143 vec3 calculateLighting(vec3 pointOnSurface, vec3 surfaceNormal, vec3 lightPosition, vec3 cameraPosition)
144 {
145     vec3 fromPointToLight = normalize(lightPosition - pointOnSurface);
146     float diffuseStrength = clamp( dot( surfaceNormal, fromPointToLight ), 0.0, 1.0 );
147
148     vec3 diffuseColor = diffuseStrength * vec3( 1.0, 0.0, 0.0 );
149     vec3 reflectedLightVector = normalize( reflect( -fromPointToLight, surfaceNormal ) );
150
151     vec3 fromPointToCamera = normalize( cameraPosition - pointOnSurface );
152     float specularStrength = pow( clamp( dot(reflectedLightVector, fromPointToCamera), 0.0, 1.0 ), 2.0 );
153
154     // Ensure that there is no specular lighting when there is no diffuse lighting.
155     specularStrength = min( diffuseStrength, specularStrength );
156     vec3 specularColor = specularStrength * vec3( 1.0 );
157
158     vec3 finalColor = diffuseColor + specularColor;
159
160     return finalColor;
161 }
162
163 //-----
164 // This is where everything starts!
165 void main( void )
166 {
167     // gl_FragCoord.xy is the coordinate of the current pixel being rendered.
168     // It is in screen space. For example if you resolution is 800x600, gl_FragCoord.xy
169     // could be (300,400). By dividing the fragcoord by the resolution, we get normalized
170     // coordinates between 0.0 and 1.0. I would like to work in a -1.0 to 1.0 space
171     // so I multiply the result by 2.0 and subtract 1.0 from it.
172     // if (gl_FragCoord.xy / resolution.xy) equals 0.0, then 0.0 * 2.0 - 1.0 = -1.0
173     // if (gl_FragCoord.xy / resolution.xy) equals 1.0, then 1.0 * 2.0 - 1.0 = 1.0
174     vec2 uv = ( gl_FragCoord.xy / resolution.xy ) * 2.0 - 1.0;
175
176     // I am assuming you have more pixels horizontally than vertically so I am multiplying
177     // the x coordinate by the aspect ratio. This means that the magnitude of x coordinate
178     // be larger than 1.0. This allows our image to not look squashed.
179     uv.x *= resolution.x / resolution.y;
180
181     // We would like to cast a ray through each pixel on the screen.
182     // In order to use a ray, we need an origin and a direction.
183     // The cameraPosition is where we want our camera to be positioned. Since our sphere will
184     // be positioned at (0,0,0), I will push our camera back by -10 units so we can see the sphere.
185     vec3 cameraPosition = vec3( 0.0, 0.0, -10.0 );
186
187     // We will need to shoot a ray from our camera's position through each pixel. To do this
188     // we will exploit the uv variable we calculated earlier, which describes the pixel we are
189     // currently rendering, and make that our direction vector.
190     vec3 cameraDirection = normalize( vec3( uv.x, uv.y, 1.0 ) );
191
192     // Now that we have our ray defined, we need to trace it to see how far the closest point
193     // in our world is to this ray. We will simply shade our scene.
194     vec3 pointOnSurface;
195     float distanceToClosestPointInScene = trace( cameraPosition, cameraDirection, pointOnSurface );
196
197     // We will now shade the sphere if our ray intersected with it.
198     vec3 finalColor = vec3(0.0);
199     if( distanceToClosestPointInScene > 0.0 )
200     {
201         vec3 lightPosition = vec3( 0.0, 4.5, -10.0 );
202         vec3 surfaceNormal = getNormal( pointOnSurface );
203         finalColor = calculateLighting( pointOnSurface, surfaceNormal, lightPosition, cameraPosition );
204     }
205
206     // And voila! We are done! We should now have a sphere! =D
207     // gl_FragColor is the final color we want to render for whatever pixel we are currently

```

```
208 |   gl_FragColor = vec4( finalColor, 1.0 );  
209 | }
```



Share It!